AD-A056 637

AFFDL-TR-78-17
Volume II

RELIABILITY-BASED SCATTER FACTORS
Volume II: Computer Manual

M. SHINOZUKA
MODERN ANALYSIS INC.
RIDGEWOOD, N. J. 07450

MARCH 1978

TECHNICAL REPORT AFFDL-TR-78-17, Volume II
Final Report March 1977 – March 1978

Approved for public release; distribution unlimited.

AIR FORCE FLIGHT DYNAMICS LABORATORY
AIR FORCE WRIGHT AERONAUTICAL LABORATORIES
AIR FORCE SYSTEMS COMMAND
WRIGHT-PATTERSON AIR FORCE BASE, OHIO 45433

NOTICE

When Government drawings, specifications, or other data are used for any purpose other than in connection with a definitely related Government procurement operation, the United States Government thereby incurs no responsibility nor any obligation whatsoever; and the fact that the Government may have formulated, furnished, or in any way supplied the said drawings, specifications, or other data, is not to be regarded by implication or otherwise as in any manner licensing the holder or any other person or corporation, or conveying any rights or permission to manufacture, use, or sell any patented invention that may in any way be related thereto.

This report has been reviewed by the Information Office (IO) and is releasable to the National Technical Information Service (NTIS). At NTIS, it will be available to the general public, including foreign nations.

This technical report has been reviewed and is approved for publication.

H. Leon Harter

H. LEON HARTER
Project Engineer

Richard D. Krobusek

RICHARD D. KROBUSEK
Major, USAF
Chief, Analysis and
Optimization Branch

FOR THE COMMANDER

Holland B. Lowndes Jr

HOLLAND B. LOWNDES, JR.
Acting Chief, Structural
Mechanics Division

Copies of this report should not be returned unless return is required by security considerations, contractual obligations, or notice on a specific document.

Unclassified

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE		READ INSTRUCTIONS BEFORE COMPLETING FORM
1. REPORT NUMBER AFFDL-TR-78-17, Volume II	2. GOVT ACCESSION NO.	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) RELIABILITY-BASED SCATTER FACTORS VOLUME II: COMPUTER MANUAL		5. TYPE OF REPORT & PERIOD COVERED Final Report March 77 - March 78
		6. PERFORMING ORG. REPORT NUMBER
7. AUTHOR(s) Masanobu Shinozuka		8. CONTRACT OR GRANT NUMBER(s) F33615-77-C-3055
9. PERFORMING ORGANIZATION NAME AND ADDRESS Modern Analysis Inc. 229 Oak Street Ridgewood, NJ 07450		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS 61102F 2304N106
11. CONTROLLING OFFICE NAME AND ADDRESS Air Force Flight Dynamics Laboratory/FBRD Air Force Systems Command Wright-Patterson AFB, OH 45433		12. REPORT DATE March 1978
		13. NUMBER OF PAGES 29
14. MONITORING AGENCY NAME & ADDRESS(if different from Controlling Office)		15. SECURITY CLASS. (of this report) Unclassified
		15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

Approved for public release; distribution unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Safety, Reliability, Scatter Factor, Weibull Distribution, Scale Parameter, Shape Parameter, Monte Carlo Simulation, Maximum Likelihood Estimates.

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

A definition of scatter factor is introduced that is rational and at the same time can directly be related to the reality of aircraft design and certification as well as of the full-scale and also coupon fatigue test of structural elements or components. Specifically, the scatter factor is defined as the ratio of the MLE (maximum likelihood estimate) of the scale parameter of the two-parameter Weibull distribution assumedly describing the life distribution of structural elements or components, to the "time to first failure" among a

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

Unclassified

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

fleet of nominally identical elements or components subjected also to nominally identical operating conditions. Freudenthal has used the same definition of the scatter factor, however, under much simplified conditions: He assumes that the shape parameter of the Weibull distribution is known. This assumption is mathematically highly convenient since it permits the derivation of the distribution of the scatter factor in closed form and independent of the unknown scale parameter. Unfortunately, however, such an assumption is inconsistent with the reality where the Weibull shape parameter easily ranges from 2.0 to 10.0 reflecting the fact that structural elements or components suffer from a variety of sources of randomness in fatigue strength; not only from the probabilistic variation of the material property but also from the statistical variation in workmanship associated with, for example, drilling rivet holes in the process of airframe fabrications. The mathematical difficulty, however, multiplies when the Weibull shape and scale parameters are both assumed to be unknown. Procedures involving Monte Carlo techniques have been established to evaluate the scatter factor under these conditions, using the maximum likelihood estimates of the parameters. The fleet reliability can then be estimated on the basis of the scatter factor thus evaluated. The effect of the sample size to be used in the fatigue test, of the fleet size and of the reliability level on the accuracy of such estimation has also been discussed.

FOREWORD

This users' manual report was prepared by Modern Analysis Inc., Ridgewood, N.J., under Air Force Contract No. F33615-77-C-3055. The work was monitored by the Air Force Flight Dynamics Laboratory, Air Force Systems Command, United States Air Force, Wright-Patterson Air Force Base, Ohio, Project 2304N106, Reliability-Based Scatter Factors, with Dr. H. Leon Harter (AFFDL/FBRD) as Project Engineer.

The project was conducted under the general direction of Dr. M. Shinozuka, President, Modern Analysis Inc. with Mr. D. Li providing the programming and documentation efforts. The programs documented herein are written for operation on a CDC 6600 computer system.

The work reported herein was performed during the period March 1977 to March 1978. The final report in two volumes was submitted on March 7, 1978.

TABLE OF CONTENTS

LIST OF ILLUSTRATIONS

LIST OF TABLES

SECTION I

INTRODUCTION

Statistical scatter factors to be used in the reliability assessment of aircraft structures were evaluated. Two programs were written for this purpose. The first program (A7701) calculates the expected value of the reliability $E[R']$ (Eq.22), the ratio $E[R']/R$ where R is the assigned value of reliability, the coefficient of variation $V_{R'} = \sigma_{R'}/E[R']$ where the $\sigma_{R'} = \{E[(R')^2] - (E[R'])^2\}^{\frac{1}{2}}$ is given in Eq. 23, and the corresponding scatter factors from Eq. 6, S based on R and S1 based on $E[R']$. The equation numbers correspond to equations in Volume I. The calculations in the program A7701 were based on the criteria that the shape parameter in the Weibull distribution is known and the scale parameter is found by maximum likelihood estimation (MLE). In the second program (A7720) the scatter factors are estimated for the case where both the scale and the shape parameters in the Weibull distribution are unknown. For this purpose the Monte Carlo simulation technique is introduced. The distributions of parameters u^*, v_0^* and Z defined in Eqs. 32, 33, 36 respectively, are empirically estimated. This program also calculates the two dimensional frequencies of u and v_0 (Eqs. 28 and 29), the reliability ratio $E[R'']/R$, and the coefficient of variation $V_{R''} = \sigma_{R''}/E[R'']$ $(\sigma_{R''} = E[(R'')^2]-$

$(E[R''])^2\}^{\frac{1}{2}})$ where $E[R'']$ and $E[(R'')^2]$ are obtained from Eq. 41. The following sections of this document discuss these two programs, the input data, and the output.

SECTION II

SCATTER FACTOR WITH KNOWN SHAPE PARAMETER (Program A7701)

A simplified flow chart of Program A7701 is shown in Figure 1.

2.1 INPUT DATA

There are no input data cards used for this program. The necessary data parameters need to be specified at the beginning of the program through DATA statements. The input parameters are as follows: AM = grid size indicating the fleet size of aircraft; A = shape parameter in the Weibull distribution; R = fleet reliability; NPT = number of division points on the distribution of Z; DZ = increment size in the distribution.

2.2 PROGRAM DESCRIPTION

Program A7701 consists of a main program unit and two subroutines: GAMMA and SIMPN. The program listing is given in Appendix A.



Figure 1. Simplified Flow Chart of Program A7701

2.2.1 The Main Program Unit (A7701)

The main program unit calculates the expected value of E[R'] of R' from Eq. 22, the ratio E[R']/R where R is given as the input parameter, the variance VAR[R'] from Eqs. 22 and 23 as $VAR[R'] = (\sigma_{R'})^2 = E[(R')^2] - (E[R'])^2$ and the scatter factors, S based on R and S1 based on E[R']. The values for the Gamma function and the numerical integration necessary to complete Eqs. 22 and 23 are obtained from subroutines GAMMA and SIMPN.

2.2.2 Subroutine GAMMA (N, GA)

Subroutine GAMMA evaluates the Gamma function $\Gamma(n) = (n-1)!$ for specified positive integer values of n. The arguments for this subroutine are:

N = n, argument of the Gamma function

GA = output value of the Gamma function

2.2.3 Subroutine SIMPN (AF, NPOINT, DSTEP, AREA)

Subroutine SIMPN is a numerical integration routine which utilizes Simpson's rule to obtain area for a specified function. The arguments for this subroutine are:

AF = function to be integrated

NPOINT = number of points or subdivisions on the function AF

DSTEP = integration step

AREA = area from the integral

2.3 PROGRAM OUTPUT

Program A7701 prints out the output as shown in Table 1.

TABLE 1. Results from Program A7701

```
SAMPLE SIZE N=     1

WEIBULL SHAPE A(  1)=    .50000       INCREMENT ON Z DZ=    .02500

FLEET RELIABILITY LEVEL R=     .50000
   E(R")   =  .42468E+00      E(R")/R=   .84937E+00
   VAR(R")=  .66335E-01      STD     =  .25756E+00
   VR"     =  .60647E+00
            FLEET SIZE        S.F. BASED ON R      S.F. BASED ON E(R")
            M=     1.         S=    .10000E+01     S1=     .54490E+00
            M=     3.         S=    .90000E+01     S1=     .49041E+01
            M=     5.         S=    .25000E+02     S1=     .13622E+02
            M=    10.         S=    .10000E+03     S1=     .54490E+02
            M=    25.         S=    .62500E+03     S1=     .34056E+03
            M=   100.         S=    .10000E+05     S1=     .54490E+04
            M=   250.         S=    .62500E+05     S1=     .34056E+05
            M=  1000.         S=    .10000E+07     S1=     .54490E+06
```

N = sample size in fatigue test; n in the text

A(1) = shape parameter α in Weibull distribution

DZ = increment size dz for numerical integration

R = fleet reliability level

E(R") = expected value of R'; E[R'] in Volume I

E(R")/R = reliability ratio; E[R']/R in Volume I

VAR(R")= variance of R'

STD = standard deviation of R'; $\sigma_{R'}$

VR" = coefficient of variation; $V_{R'}$ in Volume I

M = aircraft fleet size

S = scatter factor, S, based on R; from Eq. 6

S1 = scatter factor, S, based on E[R']; from Eq. 6

The output such as that shown in Table 1 was generated for N = 1,2,..., 10; A = 0.5, 1, 2, 3, 4, 5, 10; R = 0.5, 0.6, 0.7, 0.8, 0.9, 0.99, 0.995, 0.999, 0.9999.

SECTION III
SCATTER FACTOR WITH UNKNOWN SHAPE AND SCALE PARAMETERS
(Program A7720)

A simplified flow chart of Program A7720 is shown in Figure 2.

3.1 INPUT DATA

There are no input data cards used for this program. The parameters are specified at the beginning of the program. The input parameters are as follows: R = fleet reliability, M = fleet size, NAM = classification ASCII code for results, N = sample size in fatigue test; n in the text, NSMPL = sample size in Monte Carlo simulation; N in the text, IO = input/output peripheral device number, IX = internal parameter in RANDU subroutine , NST = a lower bound in array N, NND = an upper bound number in array N, NU = number of divisions on u, NV0 = number of divisions on v_0, UU = upper cut-off value on u, UV0 = upper cut-off value on v_0. The parameters u and v_0 are described in Volume I.

3.2 PROGRAM DESCRIPTION

Program A7720 consists of a main program and six



Figure 2. Simplified Flow Chart of Program A7720

subroutines: SAMPL, ANS, RANDU, ASCEN, LIST and DGEN. The program listing is given in Appendix B.

3.2.1 The Main Program Unit A7720

The main program unit calculates distributions of u, v_0, u^*, v_0^* and Z where these parameters are defined by Eqs. 32, 33 and 36 respectively in Volume I. The two dimensional frequencies of u and v_0 are obtained by utilizing the Monte Carlo simulation procedure. The reliability ratio $E[R'']/R$, the variance of R", $VAR = E[(R'')^2] - (E[R''])^2$, the standard deviation of R", $\sigma_{R''} = (VAR)^{1/2}$, and the coefficient of variation $V_{R''} = \sigma_{R''}/E[R'']$ are calculated from Eq. 41.

3.2.2 Subroutine SAMPL (U, V0, V0S, IX, IY, N, NSMPL, IO)

Subroutine SAMPL is used to generate values for u, v_0 and v_0^* by using the Maximum Likelihood Estimation (MLE) procedure. The arguments for this subroutine are

U = u, output parameter

V0 = v_0, output parameter

V0S = v_0^*, output parameter

IX = any odd integer less than nine digits

IY = internal parameter

N = sample size in fatigue test; n in Volume I

NSMPL = sample size in Monte Carlo simulation;
N in Volume I

IO = input/output peripheral device number

3.2.3 Subroutine ANS (A, NSMPL, NS, IO, NAM1, NAM2, M)

Subroutine ANS is used to arrange the input data in proper order and to print out the probability distribution. The arguments for this subroutine are

A = input/output array

NSMPL = sample size in Monte Carlo simulation

NS = sample size

IO = input/output peripheral number

NAM1 and NAM2 = classification parameters for the results

M = internal program parameter, or fleet size

3.2.4 Subroutine RANDU (IX, IY, RAN)

Subroutine RANDU generates independent random numbers uniformly distributed between 0 and 1. An odd integer, IX, needs to be specified in the main program. By changing IX, a new sequence of random numbers is generated. The arguments for this subroutine are

IX = any odd integer less than 9 digits

IY = internal parameter in the program

RAN = output uniform random number between 0 and 1.

3.2.5 Subroutine ASCEN (S, NPT)

Subroutine ASCEN is used to rearrange given data in ascending order. The arguments of this subroutine are

S = an arbitrary input/output array

NPT = total number of values in the array

3.2.6 Subroutine LIST (A, NPT, IO, NAM1, NAM2)

Subroutine LIST is used for listing the input as a probability distribution. The arguments of this subroutine are

A = input data arranged in ascending order

NPT = total number of values in array A

IO = output peripheral device number

NAM1 and NAM2 = classification parameters for the list

3.2.7 Subroutine DGEN (Y, N, IX, RAN)

Subroutine DGEN generates exponential random numbers by utilizing a uniform distribution in subroutine RANDU. The arguments of this subroutine are

Y = the exponential random number

N = index indicating the array of random numbers

IX = defined in RANDU

RAN = defined in RANDU

3.3 PROGRAM OUTPUT

Progam A7720 prints out the output as shown in Tables 2 - 9.

Table 2 Values for v_0^*

DISTRIBUTION OF V0* FOR SAMPLE SIZE N= 3

SAMPLE	NO.=	100	PROB=	.0500	V0*	=	.6230149E-01
SAMPLE	NO.=	200	PROB=	.1000	V0*	=	.1618284E+00
SAMPLE	NO.=	300	PROB=	.1500	V0*	=	.2759731E+00
SAMPLE	NO.=	400	PROB=	.2000	V0*	=	.3760098E+00
SAMPLE	NO.=	500	PROB=	.2500	V0*	=	.4550813E+00
SAMPLE	NO.=	600	PROB=	.3000	V0*	=	.5341889E+00
SAMPLE	NO.=	700	PROB=	.3500	V0*	=	.6099762E+00
SAMPLE	NO.=	800	PROB=	.4000	V0*	=	.6964087E+00
SAMPLE	NO.=	900	PROB=	.4500	V0*	=	.7882142E+00
SAMPLE	NO.=	1000	PROB=	.5000	V0*	=	.8803579E+00
SAMPLE	NO.=	1100	PROB=	.5500	V0*	=	.9898615E+00
SAMPLE	NO.=	1200	PROB=	.6000	V0*	=	.1090346E+01
SAMPLE	NO.=	1300	PROB=	.6500	V0*	=	.1214184E+01
SAMPLE	NO.=	1400	PROB=	.7000	V0*	=	.1353251E+01
SAMPLE	NO.=	1500	PROB=	.7500	V0*	=	.1553896E+01
SAMPLE	NO.=	1600	PROB=	.8000	V0*	=	.1838503E+01
SAMPLE	NO.=	1700	PROB=	.8500	V0*	=	.2362058E+01
SAMPLE	NO.=	1800	PROB=	.9000	V0*	=	.3148309E+01
SAMPLE	NO.=	1900	PROB=	.9500	V0*	=	.5533883E+01
SAMPLE	NO.=	1960	PROB=	.9800	V0*	=	.1462615E+02
SAMPLE	NO.=	1980	PROB=	.9900	V0*	=	.3651022E+02
SAMPLE	NO.=	1990	PROB=	.9950	V0*	=	.8491725E+03
SAMPLE	NO.=	1996	PROB=	.9980	V0*	=	.1077416E+08
SAMPLE	NO.=	1998	PROB=	.9990	V0*	=	.7369510E+11
SAMPLE	NO.=	1999	PROB=	.9995	V0*	=	.2794072E+14

V0* = v_0^*

SAMPLE NO. = a positive integer such that the plotting position of a realization is given by (SAMPLE NO.)/(NSMPL + 1)

PROB = plotting position

Table 3 Values for u*

DISTRIBUTION OF U* FOR SAMPLE SIZE N= 3

SAMPLE NO.= 100	PROB= .0500	U*	= .1752809E+00
SAMPLE NO.= 200	PROB= .1000	U*	= .2507607E+00
SAMPLE NO.= 300	PROB= .1500	U*	= .3056521E+00
SAMPLE NO.= 400	PROB= .2000	U*	= .3624199E+00
SAMPLE NO.= 500	PROB= .2500	U*	= .4218442E+00
SAMPLE NO.= 600	PROB= .3000	U*	= .4834875E+00
SAMPLE NO.= 700	PROB= .3500	U*	= .5392864E+00
SAMPLE NO.= 800	PROB= .4000	U*	= .5841478E+00
SAMPLE NO.= 900	PROB= .4500	U*	= .6247669E+00
SAMPLE NO.= 1000	PROB= .5000	U*	= .6737654E+00
SAMPLE NO.= 1100	PROB= .5500	U*	= .7317841E+00
SAMPLE NO.= 1200	PROB= .6000	U*	= .7828305E+00
SAMPLE NO.= 1300	PROB= .6500	U*	= .8405482E+00
SAMPLE NO.= 1400	PROB= .7000	U*	= .9040222E+00
SAMPLE NO.= 1500	PROB= .7500	U*	= .9801127E+00
SAMPLE NO.= 1600	PROB= .8000	U*	= .1055649E+01
SAMPLE NO.= 1700	PROB= .8500	U*	= .1151112E+01
SAMPLE NO.= 1800	PROB= .9000	U*	= .1286268E+01
SAMPLE NO.= 1900	PROB= .9500	U*	= .1554481E+01
SAMPLE NO.= 1960	PROB= .9800	U*	= .1768360E+01
SAMPLE NO.= 1980	PROB= .9900	U*	= .1902801E+01
SAMPLE NO.= 1990	PROB= .9950	U*	= .2182300E+01
SAMPLE NO.= 1996	PROB= .9980	U*	= .2475547E+01
SAMPLE NO.= 1998	PROB= .9990	U*	= .2891182E+01
SAMPLE NO.= 1999	PROB= .9995	U*	= .2940728E+01

U* = u*

SAMPLE NO. = a positive integer such that the plotting position of a realization is given by (SAMPLE NO.)/(NSMPL + 1)

PROB = plotting position

Table 4 Values for v_0

DISTRIBUTION OF V0 FOR SAMPLE SIZE N= 3

SAMPLE NO.= 100	PROB= .0500	V0	= .2622875E+00
SAMPLE NO.= 200	PROB= .1000	V0	= .3662845E+00
SAMPLE NO.= 300	PROB= .1500	V0	= .4307703E+00
SAMPLE NO.= 400	PROB= .2000	V0	= .5051354E+00
SAMPLE NO.= 500	PROB= .2500	V0	= .5794380E+00
SAMPLE NO.= 600	PROB= .3000	V0	= .6378933E+00
SAMPLE NO.= 700	PROB= .3500	V0	= .7064241E+00
SAMPLE NO.= 800	PROB= .4000	V0	= .7713822E+00
SAMPLE NO.= 900	PROB= .4500	V0	= .8326998E+00
SAMPLE NO.= 1000	PROB= .5000	V0	= .9144175E+00
SAMPLE NO.= 1100	PROB= .5500	V0	= .9893186E+00
SAMPLE NO.= 1200	PROB= .6000	V0	= .1065538E+01
SAMPLE NO.= 1300	PROB= .6500	V0	= .1138658E+01
SAMPLE NO.= 1400	PROB= .7000	V0	= .1228765E+01
SAMPLE NO.= 1500	PROB= .7500	V0	= .1328854E+01
SAMPLE NO.= 1600	PROB= .8000	V0	= .1467891E+01
SAMPLE NO.= 1700	PROB= .8500	V0	= .1600763E+01
SAMPLE NO.= 1800	PROB= .9000	V0	= .1801288E+01
SAMPLE NO.= 1900	PROB= .9500	V0	= .2160133E+01
SAMPLE NO.= 1960	PROB= .9800	V0	= .2542941E+01
SAMPLE NO.= 1980	PROB= .9900	V0	= .2943889E+01
SAMPLE NO.= 1990	PROB= .9950	V0	= .3307421E+01
SAMPLE NO.= 1996	PROB= .9980	V0	= .3659089E+01
SAMPLE NO.= 1998	PROB= .9990	V0	= .3849936E+01
SAMPLE NO.= 1999	PROB= .9995	V0	= .4230345E+01

V0 = v_0

SAMPLE NO. = a positive integer such that the plotting position of a realization is given by (SAMPLE NO.)/(NSMPL + 1)

PROB = plotting position

Table 5 Values for u

DISTRIBUTION OF U FOR SAMPLE SIZE N= 5

SAMPLE NO.=	100	PROB=	.0500	U	=	.6433014E+00
SAMPLE NO.=	200	PROB=	.1000	U	=	.7774431E+00
SAMPLE NO.=	300	PROB=	.1500	U	=	.8687249E+00
SAMPLE NO.=	400	PROB=	.2000	U	=	.9472842E+00
SAMPLE NO.=	500	PROB=	.2500	U	=	.1020291E+01
SAMPLE NO.=	600	PROB=	.3000	U	=	.1106168E+01
SAMPLE NO.=	700	PROB=	.3500	U	=	.1189700E+01
SAMPLE NO.=	800	PROB=	.4000	U	=	.1277416E+01
SAMPLE NO.=	900	PROB=	.4500	U	=	.1366523E+01
SAMPLE NO.=	1000	PROB=	.5000	U	=	.1484196E+01
SAMPLE NO.=	1100	PROB=	.5500	U	=	.1600597E+01
SAMPLE NO.=	1200	PROB=	.6000	U	=	.1711895E+01
SAMPLE NO.=	1300	PROB=	.6500	U	=	.1854302E+01
SAMPLE NO.=	1400	PROB=	.7000	U	=	.2068305E+01
SAMPLE NO.=	1500	PROB=	.7500	U	=	.2370543E+01
SAMPLE NO.=	1600	PROB=	.8000	U	=	.2759230E+01
SAMPLE NO.=	1700	PROB=	.8500	U	=	.3271693E+01
SAMPLE NO.=	1800	PROB=	.9000	U	=	.3987866E+01
SAMPLE NO.=	1900	PROB=	.9500	U	=	.5705129E+01
SAMPLE NO.=	1960	PROB=	.9800	U	=	.1001000E+02
SAMPLE NO.=	1980	PROB=	.9900	U	=	.1581000E+02
SAMPLE NO.=	1990	PROB=	.9950	U	=	.2361000E+02
SAMPLE NO.=	1996	PROB=	.9980	U	=	.4441000E+02
SAMPLE NO.=	1998	PROB=	.9990	U	=	.7980998E+02
SAMPLE NO.=	1999	PROB=	.9995	U	=	.7980998E+02

U = u

SAMPLE NO. = a positive integer such that the plotting position of a realization is given by (SAMPLE NO.)/(NSMPL + 1)

PROB = plotting position

Table 6. Joint Values of u and v_0 Outside the Specified Ranges UU or UV0

N= 3 NSMPL= 1999 IX= 25107

NU= 50 NV0= 20 UU= 25.0000 UV0= 5.0000 DU= .5000 DV0= .2500

OUTLIERS EITHER IN U OR IN V0

U=	.4441E+02	V0=	.2682E+00	J=	89	I=	2
U=	.3981E+02	V0=	.6455E+00	J=	80	I=	3
U=	.2641E+02	V0=	.1355E+01	J=	53	I=	6
U=	.3221E+02	V0=	.1653E+01	J=	65	I=	7
U=	.2641E+02	V0=	.5826E+00	J=	53	I=	3
U=	.6361E+02	V0=	.6945E+00	J=	128	I=	3
U=	.7981E+02	V0=	.1474E+01	J=	160	I=	6
U=	.7981E+02	V0=	.6705E+00	J=	160	I=	3

N = sample size in fatigue test; n in Volume I

NSMPL = sample size in Monte Carlo simulation; N in Volume I

IX = defined in RANDU

I, J = array of numbers

NU = number of divisions on u

NV0 = number of divisions on v_0

UU = upper cut-off for u

UV0 = upper cut-off for v_0

DU = increment on u, DU = UU/NU

DV0 = increment on v_0, DV0= UV0/NV0

U = u

V0 = v_0

Table 7. Joint Frequency of u and v_0, $F(u, v_o)$

F(U,V0) FOR SAMPLE SIZE N= 3
U AS COLUMN, AND V0 AS ROW.

1	6	3	1	1	1	0	0	0	0	0	0	0	0	0	0
30	104	89	82	59	48	19	16	5	4	2	0	2	1	0	0
22	64	96	102	90	61	44	17	17	14	4	0	2	2	1	0
11	46	68	54	57	33	33	21	10	11	6	3	1	0	1	0
5	25	21	29	34	18	14	12	3	6	3	0	1	0	0	0
5	11	20	9	22	12	11	8	4	3	1	1	1	0	1	1
3	13	27	14	12	4	13	5	4	1	0	0	0	0	0	0
3	6	11	12	5	8	3	3	1	3	1	0	1	0	0	0
1	4	10	10	6	3	3	2	2	0	0	1	0	0	0	0
2	11	3	4	8	1	1	0	1	0	0	0	0	0	0	0
1	1	1	3	3	3	1	1	0	2	0	0	0	0	0	0
0	1	2	5	3	2	2	1	0	0	0	0	0	0	0	0
1	2	3	3	1	2	1	0	1	0	1	0	0	0	0	0
0	1	2	2	3	1	1	0	0	0	0	0	0	0	0	0
0	0	2	2	1	0	1	0	0	0	0	0	0	1	0	0
0	0	1	2	1	0	0	1	0	0	0	0	0	0	0	0
0	1	1	2	1	0	0	0	0	0	0	0	0	0	0	0
0	0	1	1	1	0	0	0	0	0	0	0	0	0	0	0
0	0	1	0	2	1	0	0	0	0	0	1	0	0	0	0
0	0	1	0	0	0	0	0	0	0	0	0	0	0	0	0
0	0	1	1	0	0	0	1	0	0	0	0	0	0	0	0
1	0	0	2	0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	1	0	0	0	1	0	0	0	0	0	0	0
0	0	0	0	1	0	0	0	0	0	0	0	0	0	0	0
0	0	1	1	0	0	0	0	0	0	0	0	0	0	0	0
0	1	0	0	0	0	0	0	0	0	0	0	0	0	0	0
0	1	0	1	1	0	0	0	0	0	0	0	0	0	0	0
0	0	1	1	0	0	0	0	0	0	0	0	0	0	0	0
0	1	0	0	0	0	0	0	0	0	0	0	0	0	1	0
0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	1	1	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	1	0	0	0	0	0	0	0	0	0	0	0
0	3	0	0	0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
0	0	1	0	0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	2	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0	0	0	0	0
0	0	0	0	0	0	0	0	0	0	0	0	0	0	1	0

Table 8 Values for Z

DISTRIBUTION OF Z FOR FLEET SIZE M= 1

SAMPLE NO.=	100	PROB=	.0500	Z	=	-.1527202E+01
SAMPLE NO.=	200	PROB=	.1000	Z	=	-.8681633E+00
SAMPLE NO.=	300	PROB=	.1500	Z	=	-.6038523E+00
SAMPLE NO.=	400	PROB=	.2000	Z	=	-.4535508E+00
SAMPLE NO.=	500	PROB=	.2500	Z	=	-.3405964F+00
SAMPLE NO.=	600	PROB=	.3000	Z	=	-.2223126E+00
SAMPLE NO.=	700	PROB=	.3500	Z	=	-.1005370E+00
SAMPLE NO.=	800	PROB=	.4000	Z	=	-.1083520E-01
SAMPLE NO.=	900	PROB=	.4500	Z	=	.9215222E-01
SAMPLE NO.=	1000	PROB=	.5000	Z	=	.1713842E+00
SAMPLE NO.=	1100	PROB=	.5500	Z	=	.2748848E+00
SAMPLE NO.=	1200	PROB=	.6000	Z	=	.3893840E+00
SAMPLE NO.=	1300	PROB=	.6500	Z	=	.5239420E+00
SAMPLE NO.=	1400	PROB=	.7000	Z	=	.6791728E+00
SAMPLE NO.=	1500	PROB=	.7500	Z	=	.8780720E+00
SAMPLE NO.=	1600	PROB=	.8000	Z	=	.1147860E+01
SAMPLE NO.=	1700	PROB=	.8500	Z	=	.1417387E+01
SAMPLE NO.=	1800	PROB=	.9000	Z	=	.1894801E+01
SAMPLE NO.=	1900	PROB=	.9500	Z	=	.3110752E+01
SAMPLE NO.=	1960	PROB=	.9800	Z	=	.4574301E+01
SAMPLE NO.=	1980	PROB=	.9900	Z	=	.6106614E+01
SAMPLE NO.=	1990	PROB=	.9950	Z	=	.7769789E+01
SAMPLE NO.=	1996	PROB=	.9980	Z	=	.2141146E+02
SAMPLE NO.=	1998	PROB=	.9990	Z	=	.2755857E+02
SAMPLE NO.=	1999	PROB=	.9995	Z	=	.2951978E+02

Z = Z

M = aircraft fleet size

SAMPLE NO. = a positive integer such that the plotting position of a realization is given by (SAMPLE NO.)/(NSMPL + 1)

PROB = plotting position

Table 9 Expected Value, Variance, Standard Deviation, Reliability Ratio, and Coefficient of Variation of R".

N= 5 NSMPL= 1999 M= 1

R	E(R"")	VAR	STD	E(R"")/R	VR""	QS
.5000	.4878	.0444	.2107	.9757	.4318	.13947E+01
.6000	.5707	.0431	.2076	.9512	.3637	.21136E+01
.7000	.6520	.0362	.1903	.9315	.2918	.34182E+01
.8000	.7323	.0296	.1720	.9153	.2349	.64077E+01
.9000	.8173	.0174	.1320	.9082	.1616	.16770E+02
.9500	.8681	.0108	.1040	.9138	.1198	.38765E+02
.9900	.9280	.0047	.0689	.9374	.0743	.25341E+03
.9990	.9637	.0020	.0449	.9647	.0465	.49613E+04
.9999	.9784	.0013	.0356	.9785	.0364	.34388E+05

N = sample size in fatigue test; n in Volume I

NSMPL = sample size in Monte Carlo simulation; N in Volume I

M = aircraft fleet size

R = assigned reliability values

E(R"") = expected values of R" obtained from Eq. 41.

VAR = $E[(R")^2] - E^2[R"]$, variance of R" in Volume I

STD = $\sigma_{R"}$, standard deviation of R" in Volume I

E(R"")/R = reliability ratio

$V_{R""}$ = coefficient of variation of R", STD/E[R"] in Volume I

QS = Q* in Volume I

```
      PROGRAM A7701( OUTPUT, TAPE6=OUTPUT)
C     PURPOSE
C       SCATTER FACTOR CALCULATION
C       WITH KNOWN WEIBULL SHAPE PARAMETER.
C     DESCRIPTION OF THE PROGRAM
C     THIS PROGRAM CALCULATES E(R"), E(R")/R,
C       VAR(R"), VR" FOR GIVEN WEIBULL SHAPE PARAMETER A
C       SAMPLE SIZE N, FLEET SIZE AM, AND RELIABILITY LEVEL R.
C       IT ALSO CALCULATES CORRESPONDING SCATTER FACTORS
C       S BASED ON R AND S1 BASED ON E(R").
C     REMARKS
C       INTEGRATION TO GET E(R") AND E((R")**2) IS CARRIED
C       OUT BY SIMPSON"S METHOD.
C     DATE  7/16/1977
C
      DIMENSION X(1000),A(50),CV(50),STD(50),Z(1000),NN(50)
      DIMENSION XX(1000),R(50),ERR(50),ARG3(1000),ARG2(1000)
      DIMENSION AM(50),ANZA(1000),VAR(50)
C
C     ASSIGNMENT OF SPECIFIED ARRAY INPUT DATA.
C       AM= FLEET SIZE
C       A= WEIBULL SHAPE PARAMETER
C       R= SPECIFIED RELIABILITY LEVEL
C       NN= SAMPLE SIZE
      DATA AM/1.,3.,5.,10.,25.,100.,250.,1000./
      DATA A/0.5,1.,2.,3.,4.,5.,10./
      DATA R/0.5,0.6,0.7,0.8,0.9,0.99,0.995,0.999,0.9999/
      DATA NN/1,2,3,4,5,6,7,8,9,10,20/
C
C     ASSIGNMENT OF ARRAY SIZE TO BE USED (UP TO 50).
C       IN -- FOR SAMPLE SIZE NN
C       IM -- FOR FLEET SIZE AM
C       IA -- FOR WEIBULL SHAPE A
C       IR -- FOR FLEET RELIABILITY
C
      IN=11
      IM=8
      IA=7
      IR=9
C
C     TO SPECIFY OUTPUT PERIPHERAL NUMBER IO.
      IO=6
C
C     TO SPECIFY NUMBER OF INCREMENTS NPT FOR SIMPSON"S
C     NUMERICAL INTEGRATION (UP TO 1000).
C     USE NPT=400 IN THIS PROGRAM.
      NPT=400
      NPT1=NPT-1
C
C     TO CALCULATE FOR EACH SAMPLE SIZE N.
  280 DO 200 JN=1,IN
      N=NN(JN)
      AN=FLOAT(N)
      AN1=1./AN
      N2=2*N
      ANN=AN**N
      CALL GAMMA (N,GA)
```

```
C
C     TO CALCULATE FOR EACH WEIBULL SHAPE PARAMETER A.
      DO 90 K=1,IA
      DZ=0.025
      IF (K .GE. 3) DZ=0.01
      PRINT 25, N,K,A(K),DZ
   25 FORMAT(1H1/////5X,* SAMPLE SIZE N=*,I5//
     15X,* WEIBULL SHAPE A(*,I3,*)=*,F10.5,
     25X,* INCREMENT ON Z DZ=*,F10.5/)
      A1=1./A(K)
      AKN1=A(K)*ANN/GA
      ATN1=A(K)*AN-1.
C
C     TO CALCULATE FOR EACH RELIABILITY LEVEL R.
      DO 50 L=1,IR
      RR=ALOG(R(L))
      PRINT 110, R(L)
  110 FORMAT(/5X,* FLEET RELIABILITY LEVEL R=*,F10.5)
C
C     TO CALCULATE ARGUMENTS IN EQS.(22) AND (23)
C     BY USING DENSITY FUNCTION OF Z.
      DO 60 I=1,NPT
      IF (L .GT. 1) GO TO 40
      Z(I)=FLOAT(I)*DZ
      ARG2(I)=-AN*Z(I)**A(K)
      IF (ARG2(I) .LT. -100.) GO TO 40
      ARG3(I)=AKN1*Z(I)**ATN1*EXP(ARG2(I))
      ANZA(I)=AN*Z(I)**A(K)
   40 IF (ARG2(I) .LT. -100.) GO TO 30
      ARG=1./(1.-RR/ANZA(I))
   80 X(I)=ARG**N*ARG3(I)
      XX(I)=X(I)*ARG**N
      GO TO 60
   30 X(I)=0.
      XX(I)=0.
   60 CONTINUE
C
C     TO PERFORM INTEGRATION TO ESTIMATE E(R") AND E((R")**2).
      CALL SIMPN(X,NPT1,DZ,AREA)
      CALL SIMPN(XX,NPT1,DZ,AREX)
C
C     TO CALCULATE E(R")/R, VAR(R"), STD(=STANDARD DEVIATION),
C     AND VR"(=COEFFICIENT OF VARIATION) AND ALSO PRINT OUT.
      ERR(L)=AREA/R(L)
      VAR(L)=AREX-AREA**2
      IF(VAR(L).LT.0.) VAR(L)=0.
      STD(L)=SQRT(VAR(L))
      CV(L)=STD(L)/AREA
      PRINT 10, AREA,ERR(L),VAR(L),STD(L),CV(L)
   10 FORMAT(9X,*E(R")   =*,E12.5,5X,*E(R")/R=*,E12.5/
     19X,*VAR(R")=*,E12.5,5X,*STD     =*,E12.5/
     29X,*VR"     =*,E12.5/
     318X,*FLEET SIZE*,5X,*S.F. BASED ON R*,4X,
     4*S.F. BASED ON E(R")*)
C
C     TO CALCULATE SCATTER FACTORS S BASED ON R
C     AND S1 BASED ON E(R") FOR EACH FLEET SIZE.
```

```
      DO 70 J=1,IM
      S1=(AM(J)/(AN*(1./AREA**AN1-1.)))**A1
      SS=(AM(J)/(AN*(1./R(L)**AN1-1.)))**A1
      PRINT 120, AM(J),SS,S1
  120 FORMAT(17X,* M=*,F6.0,7X,*S= *,E12.5,4X,*S1= *,E12.5)
   70 CONTINUE
   50 CONTINUE
   90 CONTINUE
  200 CONTINUE
      STOP
      END

      SUBROUTINE GAMMA(N,GA)
C     TO CALCULATE GAMMA FUNCTION.
C     USE FACTORIAL RELATIONSHIP SINCE N IS
C     AN INTEGER IN EVERY CASE.
C     N MUST BE LESS THAN 50.
C
      GA=1.0
      IF(N.EQ.1) RETURN
      N1=N-1
      DO 10 I=1,N1
   10 GA=GA*FLOAT(I)
      RETURN
      END

      SUBROUTINE SIMPN (AF,NPOINT,DSTEP,AREA)
      DIMENSION AF(1)
C
C     TO CARRY OUT INTEGRATION FOR GIVEN VALUES OF
C     FUNCTION AF.
      NP=(NPOINT-1)/2
      MN=NP-1
      ODD=0.
      EVEN=0.
      END=AF(1)+AF(NPOINT)
      DO 10 I=1,NP
   10 ODD=ODD+AF(2*I)
      DO 20 I=1,MN
   20 EVEN=EVEN+AF(2*I+1)
      AREA=(4.0*ODD+2.0*EVEN+END)*DSTEP/3.0
      RETURN
      END
```

```
      PROGRAM A7720 (OUTPUT, TAPE6=OUTPUT)
C     PURPOSE
C       SCATTER FACTOR CALCULATION
C       WITH UNKNOWN WEIBULL SHAPE AND SCALE PARAMETERS.
C     DESCRIPTION OF THE PROGRAM
C       THIS PROGRAM FIRST GENERATES NSMPL NUMBER OF
C       SAMPLE SETS OF SIZE N FOR U, U*, V0 AND V0*,
C       AND THEN CALCULATES THEIR PROBABILITY DISTRIBUTION
C       AS WELL AS JOINT DENSITY F(U,V0).
C       SECOND, IT CALCULATES PROBABILITY DISTRIBUTION
C       OF Z FOR SAMPLE SIZE N AND FLEET SIZE M BY GENERATING
C       CORRESPONDING SAMPLES OF W.
C       IT FINALLY CALCULATES E(R"), VAR(R"), VR", E(R")/R AND Q*.
C     REMARKS
C       IN THE PROGRAM, "*" IS REPLACED BY "S", EXAMPLE, U*=US.
C     DATE 10/25/77
C
      DIMENSION U(1999),V0(1999),V0S(1999),Z(1999)
      DIMENSION NV0U(20,50),R(9),M(8),N(10),NAM(6)
      EQUIVALENCE (V0S,Z)
C
C     ASSIGNMENT OF SPECIFIED ARRAY INPUT DATA.
C       N= SAMPLE SIZE.
C       M= FLEET SIZE.
C       R= SPECIFIED RELIABILITY LEVEL.
C       NAM= DATA CLASSIFICATION ASCII CODE.
      DATA N/2,3,4,5,6,7,8,9,10,20/
      DATA M/1,3,5,10,25,100,250,1000/
      DATA R/0.5,0.6,0.7,0.8,0.9,0.95,0.99,0.999,0.9999/
      DATA NAM/2HU ,2H  ,2HU*,2HV0,2H* ,2HZ /
C
C     INPUT DATA ASSIGNMENT.
C       NSMPL= NUMBER OF SAMPLE SETS OF SIZE N.
C       IX= INITIAL VALUES FOR UNIFORM RANDOM NUMBER
C                  GENERATION SUBROUTINE RANDU(IX,IY,RAN).
C       NST AND NND= CONTROL PARAMETER FOR DO-LOOP CALCULATION
C                  FOR EACH SAMPLE SIZE N.
C         NST= LOWER BOUND NUMBER IN ARRAY N.
C         NND= UPPER BOUND NUMBER IN ARRAY N.
C       NU= ARRAY SIZE FOR U.
C       NV0= ARRAY SIZE FOR V0.
C       UU= UPPER CUT-OFF VALUE FOR U.
C       UV0= UPPER CUT-OFF VALUE FOR V0.
C       LU AND IO= SELECT CODES OF DATA INPUT/OUTPUT PERIPHERAL.
C
      NSMPL=1999
      IX=23451
      NST=1
      NND=10
      NU=50
      NV0=20
      UU=25.
      UV0=5.
C
C     CALCULATION FOR EACH SAMPLE SIZE N(KN)
      FNSMP1=FLOAT(NSMPL+1)
      FNSMP2=1.0/FNSMP1
```

```
      DO 100 KN=NST,NND
      NS=N(KN)
C
C     TO GENERATE NSMPL NUMBER OF SAMPLE SETS OF SIZE N
C     FOR U, V0, AND V0*.
      CALL SAMPL(U,V0,V0S,IX,NS,NSMPL,IO)
C
C     TO CALCULATE PROBABILITY DISTRIBUTION OF V0*
C     BY ARRANGING IN ASCENDING ORDER.
      CALL ANS(V0S,NSMPL,NS,IO,NAM(4),NAM(5),0)
C
C     TO CALCULATE PROBABILITY DISTRIBUTION OF U*.
C     BECAUSE OF LIMITTED MEMORY SIZE OF THE COMPUTER USED,
C     AND BECAUSE OF VALUES U AND V0 TO BE RESERVED FOR
C     CALCULATION OF Z, ARRAY V0S IS USED AS COMMON ARRAY.
      DO 110 I=1,NSMPL
  110 V0S(I)=1./U(I)
      CALL ANS(V0S,NSMPL,NS,IO,NAM(3),NAM(2),0)
C
C     TO CALCULATE PROBABILITY DISTRIBUTION OF V0.
      DO 120 I=1,NSMPL
  120 V0S(I)=V0(I)
      CALL ANS(V0S,NSMPL,NS,IO,NAM(4),NAM(2),0)
C
C     TO CALCULATE PROBABILITY DISTRIBUTION OF U.
      DO 130 I=1,NSMPL
  130 V0S(I)=U(I)
      CALL ANS(V0S ,NSMPL,NS,IO,NAM(1),NAM(2),0)
C
C     TO CALCULATE JOINT DENSITY F(U,V0), E(R"), VAR(R"),
C     VR", E(R")/R, AND Q*.
C
      DU=UU/FLOAT(NU)
      DV0=UV0/FLOAT(NV0)
      DO 150 J=1,NU
      DO 150 I=1,NV0
      NV0U(I,J)=0
  150 CONTINUE
C
C     TO PRINT OUT NECESSARY INFORMATIONS FOR CHECK PURPOSE
      WRITE(6,600) NS,NSMPL,IX
  600 FORMAT(1H1/////5X,* N=*,I5,*   NSMPL=*,I5,*   IX=*,I20)
      WRITE(6,610) NU,NV0,UU,UV0,DU,DV0
  610 FORMAT(/5X,* NU=*,I5,* NV0=*,I5,* UU=*,F8.4,* UV0=*,
     1F8.4,* DU=*,F8.4,* DV0=*,F8.4//)
      WRITE(6,611)
  611 FORMAT(/5X,* OUTLIERS EITHER IN U OR IN V0*/)
      DO 160 K=1,NSMPL
      J=IFIX(U(K)/DU)+1
      I=IFIX(V0(K)/DV0)+1
      IF ( J.LE.NU .AND. I.LE.NV0) GO TO 10
C
C     TO PRINT OUT OUTLIERS EITHER IN U OR V0
C     FROM THE PRE-DEFINED RANGE.
      WRITE(6,630) U(K),V0(K),J,I
  630 FORMAT( 5X,* U=*,E12.4,3X,* V0=*,E12.4,3X,* J=*,I6,3X,
     1* I=*,I6)
```

```
      IF(J.GT.NU) J=NU
      IF(I.GT.NVO) I=NVO
   10 NVOU(I,J)=NVOU(I,J)+1
  160 CONTINUE
C     TO PRINT OUT JOINT FREQUENCY F(U,V0).
C     JOINT FREQUENCY F(U,V0) IS LISTED WITH U AS COLUMN,
C     AND V0 AS ROW.
      WRITE(6,640) NS
  640 FORMAT(1H1/////5X,*   F(U,V0) FOR SAMPLE SIZE N=*,I5/
     18X,* U AS COLUMN, AND V0 AS ROW.*/)
      WRITE(6,650) NVOU
  650 FORMAT(20(1X,I3))
C
C     FOR BRIEF VERSION OF LISTING OF F(U,V0).
      WRITE(6,640) NS
      DO 170 J=1,40
      WRITE(6,660) ( NVOU(I,J),I=1,16)
  660 FORMAT(6X,16(1X,I3))
  170 CONTINUE
C
C     TO CALCULATE DISTRIBUTION OF Z FOR EACH FLEET SIZE.
C     SAME INITIAL VALUES FOR UNIFORM RANDOM NUMBER
C     GENERATION SHALL BE USED FOR EACH CASE OF FLEET SIZE.
      IXS=IX
      DO 200 KK=1,8
      MM=M(KK)
      FM=FLOAT(MM)
      FNS=FLOAT(NSMPL)+1.
      IX=IXS
      DO 210 I=1,NSMPL
      CALL RANDU (IX,IY,RAN)
      IX=IY
      W=-ALOG(RAN)/FM
      VOW=V0(I)/W
      IF (VOW .LT. 1.E-10) VOW=1.E-10
      Z(I)=U(I)*ALOG10(VOW)
  210 CONTINUE
C
C     CALCULATE DISTRIBUTION OF Z FOR EACH FLEET SIZE
      CALL ANS(Z,NSMPL,NS,IO,NAM(6),NAM(2),MM)
C
C     TO DEFINE THE CASE CALCULATED.
      WRITE(6,670) NS,NSMPL,MM
  670 FORMAT(1H1///5X,* N=*,I5,*  NSMPL=*,I6,*  M=*,I6)
      WRITE(6,680)
  680 FORMAT(//8X,* R*,7X,* E(R"*,*")*,3X,* VAR*,6X,
     1* STD*,3X,* E(R"*,*")/R*,1X,* VR"*,*"*,7X,* QS*/)
C
C     TO ESTIMATE E(R"), VAR(R"), VR", E(R")/R, AND Q*.
      DO 300 K=1,9
      FMALR=FM/ALOG(1./R(K))
      ER=0.0
      VAR=0.0
      DU2=DU/2.
      DV02=DV0/2.
      DO 310 J=1,NU
      DO 320 I=1,NV0
```

```
      IF (NVOU(I,J) .EQ. 0) GO TO 320
      UX=FLOAT(J)*DU-DU2
      VOX=FLOAT(I)*DV0-DV02
      PARM=VOX*FMALR
      Z0=UX*ALOG10(PARM)
      PAR2=FLOAT(NVOU(I,J))/FNS
      DO 330 L=1,NSMPL
      IF (Z0 .LE. Z(L)) GO TO 20
  330 CONTINUE
   20 FZ=FLOAT(L)/FNS
      FZ2=FZ**2
      ER=ER+FZ*PAR2
      VAR=VAR+FZ2*PAR2
  320 CONTINUE
  310 CONTINUE
      VAR=VAR-ER**2
      STD=SQRT(VAR)
      CV=STD/ER
      RATIO=ER/R(K)
      ZER=ER*FNSMP1
      IZ=IFIX(ZER)
      IZ1=IZ+1
      FIZ=FLOAT(IZ)/FNSMP1
      ZANS=Z(IZ)+(ER-FIZ)*(Z(IZ1)-Z(IZ))/FNSMP2
      QS=10.**ZANS
      WRITE(6,690) R(K),ER,VAR,STD,RATIO,CV,QS
  690 FORMAT(6X,6F9.4,1X,E12.5)
  300 CONTINUE
  200 CONTINUE
  100 CONTINUE
      STOP
      END

      SUBROUTINE ANS(A,NSMPL,NS,IO,NAM1,NAM2,M)
      DIMENSION A(1)
C     TO PRINT OUT PROBABILITY DISTRIBUTION
C     FOR GIVEN ARRAY A.
      CALL ASCEN (A,NSMPL)
      IF (M .EQ. 0) GO TO 10
      WRITE(6,100) NAM1,NAM2,M
  100 FORMAT(1H1/////5X,* DISTRIBUTION OF *,2A2,
     1* FOR FLEET SIZE M=*,I6//)
      GO TO 20
   10 WRITE(6,110) NAM1,NAM2,NS
  110 FORMAT(1H1/////5X,* DISTRIBUTION OF *,2A2,
     1* FOR SAMPLE SIZE N=*,I5//)
   20 CALL LIST (A,NSMPL,IO,NAM1,NAM2)
      RETURN
      END
```

```
      SUBROUTINE LIST(A,NPT,IO,NAM1,NAM2)
      DIMENSION A(1),P(6)
C
C     TO LIST CORRESPONDING REALIZATION VALUES FOR
C     SPECIFIC PROBABILITY LEVELS.
C
      DATA P/0.98,0.99,0.995,0.998,0.999,0.9995/
      NPT1=NPT+1
      ANPT1=FLOAT(NPT1)
      ND=(NPT+1)/20
      IF (ND .LT. 1) ND=1
      DO 10 I=ND,NPT,ND
      PROB=FLOAT(I)/ANPT1
      WRITE(6,100) I,PROB,NAM1,NAM2,A(I)
   10 CONTINUE
      DO 20 I=1,6
      J=IFIX(ANPT1*P(I))
      WRITE(6,100) J,P(I),NAM1,NAM2,A(J)
      IF(J.EQ.NPT) GO TO 30
   20 CONTINUE
  100 FORMAT(5X,* SAMPLE NO.=*,I5,3X,* PROB=*,F7.4,3X,
     12A2,* =*,E14.7)
   30 RETURN
      END

      SUBROUTINE ASCEN(S,NPT)
      DIMENSION S(1)
C
C     TO ARRANGE GIVEN ARRAY S IN ASCENDING ORDER
      IF (NPT .LT. 2) RETURN
      N2=NPT/2
      DO 20 J=1,N2
      MM=NPT-J+1
      K=J
      M=MM
      DO 10 I=J,MM
      IF (S(I) .LT. S(K)) K=I
      IF (S(I) .GT. S(M)) M=I
   10 CONTINUE
C     TEST FOR J=K
      IF (J .NE. K) GO TO 50
      IF (MM .EQ. M) GO TO 20
      TEMP=S(M)
      S(M)=S(MM)
      S(MM)=TEMP
      GO TO 20
C     TEST FOR MM=M
   50 IF (MM .NE. M) GO TO 60
      TEMP=S(J)
      S(J)=S(K)
      S(K)=TEMP
      GO TO 20
   60 IF (J .EQ. M) GO TO 30
C     J/=M,MM/=K
      TEMP=S(J)
```

```
      S(J)=S(K)
      S(K)=TEMP
      TEMP1=S(MM)
      S(MM)=S(M)
      S(M)=TEMP1
      GO TO 20
   30 IF(MM .EQ. K) GO TO 40
C     J=M
      TEMP1=S(MM)
      S(MM)=S(M)
      S(M)=TEMP1
      TEMP=S(J)
      S(J)=S(K)
      S(K)=TEMP
      GO TO 20
C     J=M,MM=K
   40 TEMP=S(J)
      S(J)=S(MM)
      S(MM)=TEMP
   20 CONTINUE
      RETURN
      END

      SUBROUTINE SAMPL(U,V0,V0S,IX,N,NSMPL,IO)
      DIMENSION U(1),V0(1),V0S(1),Y(20),RAN(20),ALY(20)
C
C     TO GENERATE NSMPL NUMBER OF SAMPLE SETS OF SIZE N
C     FOR U, V0 AND V0*.
C     SAMPLE SIZE N MUST BE LESS THAN OR EQUAL TO 20 FOR
C     MINI-COMPUTER.
      DU=0.2
      NPT=400
      DO 100 L=1,NSMPL
      CALL DGEN(Y,N,IX,RAN)
      KASE=0
      PAR5=0.
      DO 110 I=1,N
      ALY(I)=ALOG(Y(I))
      PAR5=PAR5+ALY(I)
  110 CONTINUE
      PAR5=PAR5/FLOAT(N)
      UMIN=0.01
C
C     FIND THE SOLUTION OF PHI(U) IN MLE EQUATION.
      DO 120 J=1,NPT
      U1=FLOAT(J-1)*DU+UMIN
   10 PAR1=0.
      PAR2=0.
      DO 130 I=1,N
      YIU=Y(I)**U1
      PAR1=PAR1+YIU*ALY(I)
      PAR2=PAR2+YIU
  130 CONTINUE
      PHI2=PAR1/PAR2-1./U1-PAR5
      IF (ABS(PHI2).LT. 0.001) GO TO 20
      IF (J .EQ. 1) GO TO 30
      IF (KASE .EQ. 1) GO TO 20
      IF (PHI1 .GT. 0.) GO TO 40
      IF (PHI2 .LT. 0.) GO TO 30
```

```
      GO TO 50
   40 IF (PHI2 .GT. 0.) GO TO 30
   50 U1=U1-PHI2/(PHI2-PHI1)*DU
      KASE=1
      GO TO 10
C
   30 PHI1=PHI2
  120 CONTINUE
   20 U(L)=U1
      US=1./U1
      VOS(L)=PAR2/FLOAT(N)
      UV0=US*ALOG10(VOS(L))
      IF (UV0 .GT. -30.) GO TO 60
      V0(L)=0.
      GO TO 100
   60 IF (UV0 .LT. 30.) GO TO 70
      V0(L)=1.E30
      GO TO 100
   70 V0(L)=VOS(L)**US
C
  100 CONTINUE
      RETURN
      END

      SUBROUTINE DGEN(Y,N,IX,RAN)
      DIMENSION Y(1),RAN(1)
C
C     SUBROUTINE TO GENERATE EXPONENTIAL RANDOM NUMBER
C     OF SIZE N
C     BY Y=-LN(X) , WHERE X -- UNIFORM RANDOM NUMBER
C
      DO 10 I=1,N
      CALL RANDU(IX,IY,RAN(I))
      IX=IY
   10 Y(I)=-ALOG(RAN(I))
      RETURN
      END

      SUBROUTINE RANDU(IX,IY,RAN)
      M=2**32
      K=2**7+1
      IY=MOD(IX*K+1,M)
      RAN=FLOAT(IY)/FLOAT(M)
      RETURN
      END
```


☆U.S.Government Printing Office: 1978 — 757-080/668